ODS GRAPHICS DESIGNER (Creating Templates for Batchable Graphs)

Toronto Area SAS Society
March, 2012

Barry Hong
BYHong@uss.com

History of SAS Graphics

- ***In the beginning there was PROC PLOT***
  Crude raster graphics in the days of line printers

- ***Then there was SAS/GRAPH and it was better***
  Vector graphics produced quality output
  AXIS, FOOTNOTE, GOPTIONS, LEGEND, PATTERN, SYMBOL, TITLE
  Lots of options but too many to learn effectively
  Output stored in graphics catalogs
  Not too friendly with Microsoft Office products

- ***SG Graphics***
  Output as PNG file for sharing with Microsoft Office products.
  Still code driven but using a new language employing styles

- ***Graphics Template Language***
  Quality graphics fully compatible with Word and PowerPoint

ODS Graphics Designer

- ***Eliminates the need to create template styles for graphics***
  Who really mastered PROC TEMPLATE?

- ***Drag & Drop and Point & Click version of SG Graphics***
  Let ODS Graphics Designer write the code for you
  Customize the appearance to meet corporate standards

- ***Create Custom Designed Graphics***
  Can layer charts or create panels in one file.
  Make those Excel lovers jealous!

- ***Save the Template for Reuse or Sharing***
  PROC SGRENDER processes data through the template

- ***Preproduction in SAS 9.2 Release 2 (TS2M0)***
- ***Part of Base SAS in 9.3 (No need for SAS/GRAPH)***

ODS Graphics Designer - Example Output

Combination bar chart by month and by week



ODS Graphics Designer - Example Output

Combination scatter plot and histogram for the same data.



ODS Graphics Designer - Example Output

Combination box plot by month and 3 line charts overlaid by day for the past 30 days.

Let's build the template in ODS Graphics Designer

%sgdesign;



ODS Graphics Designer

Complex templates may require a common data table

- ***Chart_1_data (for box plot)***

Month	*Flow*
JUN10	*36.3*
JUN10	*32.4*
…	*…*
MAY11	*36.5*

```
DATA data.Chart_Data;
  set chart_1_data chart_2_data;
RUN;
```

- ***Chart_2_data (for line charts)***

Date	*Min*	*Mean*	*Max*
11APR1123	*36*	*40*	
12APR11	*30*	*30*	*30*
…	*…*	*…*	*…*
10MAY11	*12*	*38*	*41*

ODS Graphics Designer – Graph Gallery

Have It Your Way, Select a Graph from the Gallery



ODS Graphics Designer – Graph Gallery

The Graph Gallery contains a tabbed set of commonly used graphs, organized as follows:

- **Basic** – Common graphs
- **Grouped** – Graphs showing grouped data
- **Analytical** – Graphs commonly used for analysis of data
- **Custom** – A set of graphs showing the possible ways to combine the plots
- **Matrix** – A set of Scatter Plot Matrix graphs
- **Panels** – A set of Classification Panel Graphs
- **MyGraphs** – A user defined group

ODS Graphics Designer – Assign Data

When you select a graph the Assign Data dialog box opens.

Select the data table and the variables for the vertical box plot.



ODS Graphics Designer

Default settings for the vertical box plot.

To create a second chart: Right mouse click on the chart and select Add Column.



ODS Graphics Designer – Plot Layers

Drag and drop a Series chart from the Plot Layers onto the new plot space.



ODS Graphics Designer – Plot Layers

The same data table for the box plot is used for the Series plot. Assign DATE and MEAN to the X & Y variables.



ODS Graphics Designer – Plot Layers

Repeat the Series Plot Layer drag and drop process on top of the Series Plot to add the MIN and MAX.



ODS Graphics Designer – Plot Layers

Now that we have created the basic design it is time to customize the appearance.

Select a chart component and right mouse click to change the properties.



ODS Graphics Designer – Customization

Select either chart, bring up the Graph Properties and select Common Row Axis for both charts to use the same Y axis scaling.



ODS Graphics Designer – Customize Plot Properties

Unselect Outline for each of the plots.

For each of the series select a colour and make the line heavier Thickness=2





ODS Graphics Designer – Customize Plot Properties

Axes Tab permits colour, font and size changes to values and labels. Make changes to each axis separately.





Display Tab allows you to turn Label, Values, Grid and Tick Marks off and on.

Albany AMT is the new Arial.

Font size unit is Pts

ODS Graphics Designer – Customize

The box plot was widened by dragging the right border of the plot.

Title, ylabel & footnote made generic.

Next we want to drag and drop a legend onto the Series chart.



ODS Graphics Designer – Customize

We will move the legend when we look at the code.

ODS Graphics Designer is built with only a subset of the Template code



Now to use the PROC TEMPLATE Code



ODS Graphics Designer – Understanding the Code

Select View > Code. In the view window copy all of the code and paste it into the SAS Editor.

We will modify the code to create a template for future use.

```
proc template;
  define statgraph sgdesign;
  dynamic _FLOW _MONTH _DATE _DATE2 _MIN _DATE3 _MAX _MEAN;
  begingraph / designheight=480 designwidth=720;
    entrytitle _id='title' halign=center '_TITLE' / textattrs=(size=14 family='Albany AMT');
    entryfootnote _id='footnote' halign=left '_FOOTNOTE' /;
      layout lattice _id='lattice' / columndatarange=data columngutter=10 columnweights=(0.5252365930599369
                0.47476340694006314 ) columns=2 rowdatarange=union rowgutter=10;
        layout overlay _id='overlay' / walldisplay=(FILL) xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL
                size=10 family='Albany AMT') display=(TICKS TICKVALUES LINE ));
          boxplot _id='box' x=_MONTH y=_FLOW / name='box';
        endlayout;
        layout overlay _id='overlay2' / walldisplay=(FILL) xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL
                size=10 family='Albany AMT') display=(TICKS TICKVALUES LINE ));
          seriesplot _id='series' x=_DATE y=_MEAN / connectorder=xaxis lineattrs=(thickness=2)  name='series';
          seriesplot _id='series2' x=_DATE2 y=_MIN / connectorder=xaxis lineattrs=(color=CX9C3418
                thickness=2)    name='series2';
          seriesplot _id='series3' x=_DATE3 y=_MAX / connectorder=xaxis lineattrs=(color=CX0000FF
                thickness=2)    name='series3';
          discretelegend _id='legend' 'series' 'series2' 'series3'  / border=true displayclipped=true down=1
                halign=center   location=inside opaque=false order=columnmajor valign=bottom;
        endlayout;
        rowaxes;
          rowaxis _id='rowaxis' / label='_YLABEL' labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')
                tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT');
        endrowaxes;
      endlayout;
    endgraph;
  end;
run;
```



ODS Graphics Designer – Understanding the Code

Change name

Dynamic like macro arguments: all start with underscore.

Consolidate and make generic.

Remove quotes from _TITLE and _FOOTNOTE (now dynamic variables)

```
proc template;
  define statgraph sgdesign;
  dynamic _FLOW _MONTH _DATE _DATE2 _MIN _DATE3 _MAX _MEAN;
  begingraph / designheight=480 designwidth=720;
    entrytitle _id='title' halign=center '_TITLE' / textattrs=(size=14 family='Albany AMT');
    entryfootnote _id='footnote' halign=left '_FOOTNOTE' /;
```

Modified code

```
proc template;
  define statgraph SG_GHSUG;

 dynamic _YVARBOX _MONTH _DATE  _MIN  _MAX _MEAN _TITLE _FOOTNOTE
                 _YLABEL;

   begingraph / designheight=480 designwidth=720;
      entrytitle _id='title' halign=center _TITLE / textattrs=(size=14 family='Albany AMT');
      entryfootnote _id='footnote' halign=left _FOOTNOTE /;
```

ODS Graphics Designer – Understanding the Code

Layout lattice defines layout of charts.

Simplify columnweight and reduce column gutter

Layout overlay for the boxplot. Replace boxplot Y variable _FLOW with dynamic _YVARBOX

```
layout lattice _id='lattice' / columndatarange=data columngutter=10
  columnweights=(0.5252365930599369       0.47476340694006314 ) columns=2
  rowdatarange=union rowgutter=10;

  layout overlay _id='overlay' / walldisplay=(FILL)
     xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10
                    family='Albany AMT') display=(TICKS TICKVALUES LINE ));
      boxplot _id='box' x=_MONTH y=_FLOW / name='box';
   endlayout;
```

Modified code

```
layout lattice _id='lattice' / columndatarange=data columngutter=5
  columnweights=(0.55  0.45 ) columns=2
  rowdatarange=union rowgutter=10;

  layout overlay _id='overlay' / walldisplay=(FILL)
     xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10
                    family='Albany AMT') display=(TICKS TICKVALUES LINE ));
      boxplot _id='box' x=_MONTH y=_YVARBOX / name='box';
   endlayout;
```

ODS Graphics Designer – Understanding the Code

Replace _DATE2 and _DATE3 with common dynamic _DATE.

Rename 'series' as 'series1'

Move legend location from inside to outside.

```
layout overlay _id='overlay2' / walldisplay=(FILL)
  xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT')
     display=(TICKS TICKVALUES LINE ));
  seriesplot _id='series' x=_DATE y=_MEAN / connectorder=xaxis
     lineattrs=(thickness=2) name='series';
  seriesplot _id='series2' x=_DATE2 y=_MIN / connectorder=xaxis
     lineattrs=(color=CX9C3418      thickness=2)              name='series2';
  seriesplot _id='series3' x=_DATE3 y=_MAX / connectorder=xaxis
     lineattrs=(color=CX0000FF      thickness=2)              name='series3';
  discretelegend _id='legend' 'series' 'series2' 'series3'  / border=true displayclipped=true
     down=1  halign=center  location=inside opaque=false order=columnmajor
     valign=bottom;
endlayout;
```

Modified code

```
layout overlay _id='overlay2' / walldisplay=(FILL)
  xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT')
     display=(TICKS TICKVALUES LINE ));
  seriesplot _id='series1' x=_DATE y=_MEAN / connectorder=xaxis
     lineattrs=(thickness=2) name='series1';
  seriesplot _id='series2' x=_DATE y=_MIN / connectorder=xaxis
     lineattrs=(color=CX9C3418      thickness=2)              name='series2';
  seriesplot _id='series3' x=_DATE y=_MAX / connectorder=xaxis
     lineattrs=(color=CX0000FF      thickness=2)              name='series3';
  discretelegend _id='legend' 'series1' 'series2' 'series3'  / border=true
     displayclipped=true    down=1  halign=center  location=outside opaque=false
     order=columnmajor   valign=bottom;
endlayout;
```

ODS Graphics Designer – Understanding the Code

Remove quotes from _YLABEL (now a dynamic variable)

```
rowaxes;
   rowaxis _id='rowaxis' / label='_YLABEL‘
      labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')
      tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT');
endrowaxes;
```

Modified code

```
rowaxes;
   rowaxis _id='rowaxis' / label=_YLABEL
      labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')
      tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT');
endrowaxes;
```

ODS Graphics Designer – Understanding the Code

The PROC does not identify a data table!

It is a generic template.

Submit the code.

```
proc template;
  define statgraph SG_GHSUG;
  dynamic _YVARBOX _MONTH _DATE  _MIN  _MAX _MEAN _TITLE _FOOTNOTE _YLABEL;
  begingraph / designheight=480 designwidth=720;
    entrytitle _id='title' halign=center _TITLE / textattrs=(size=14 family='Albany AMT');
    entryfootnote _id='footnote' halign=left _FOOTNOTE /;
      layout lattice _id='lattice' / columndatarange=data columngutter=5 columnweights=(0.55  0.45 ) columns=2
                    rowdatarange=union rowgutter=10;
        layout overlay _id='overlay' / walldisplay=(FILL) xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL
                size=10 family='Albany AMT') display=(TICKS TICKVALUES LINE ));
          boxplot _id='box' x=_MONTH y=_YVARBOX / name='box';
        endlayout;
        layout overlay _id='overlay2' / walldisplay=(FILL) xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL
                size=10 family='Albany AMT') display=(TICKS TICKVALUES LINE ));
          seriesplot _id='series1' x=_DATE y=_MEAN / connectorder=xaxis lineattrs=(thickness=2) name='series1';
          seriesplot _id='series2' x=_DATE y=_MIN / connectorder=xaxis lineattrs=(color=CX9C3418
                thickness=2)    name='series2';
          seriesplot _id='series3' x=_DATE y=_MAX / connectorder=xaxis lineattrs=(color=CX0000FF
                thickness=2)    name='series3';
          discretelegend _id='legend' 'series1' 'series2' 'series3'  / border=true displayclipped=true down=1
                halign=center   location=outside opaque=false order=columnmajor valign=bottom;
        endlayout;
        rowaxes;
          rowaxis _id='rowaxis' / label=_YLABEL labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')
                tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT');
        endrowaxes;
      endlayout;
    endgraph;
  end;
run;
```

ODS Graphics Designer – Creating the Template

STATGRAPH template is saved locally in a SAS itemstore.

Want to save templates to an itemstore to share with others.

ODS path statement will search SG_TEMPLATE first.

```
NOTE: STATGRAPH 'Sg_ghsug' has been saved to: SASUSER.TEMPLAT
24    run;
NOTE: PROCEDURE TEMPLATE used (Total process time):
      real time          0.60 seconds
      cpu time           0.12 seconds
```

```
proc template;
  define statgraph sg_ghsug /store=libref.SG_TEMPLATE;
```

```
ODS path (prepend) libref.SG_TEMPLATE (read);
```

```
ods path show;
Current ODS PATH list is:

1. LIBREF.SG_TEMPLATE(READ)
2. SASUSER.TEMPLAT(UPDATE)
3. SASHELP.TMPLMST(READ)
```

Add statement to autoexec file.

ODS Graphics Designer – Using the Template

ODS LISTING specifies the output location.

PROC SGRENDER acts like a macro invocation to process the data.

Output created as a PNG file.

```
ods listing gpath="e:\temporary" image_dpi=100;

proc sgrender data=data.chart_data template=sg_ghsug
                    object=ghsug;
  dynamic _DATE="date" _MONTH="month"
          _MEAN="mean" _MAX="max" _MIN="min"
          _YVARBOX="flow" _YLABEL="L/min"
          _TITLE="Flow" _FOOTNOTE="My SG chart";
run;

ods listing;
```

```
NOTE: Listing image output written to e:\temporary\ghsug1.png.
NOTE: There were 326 observations read from the data set
DATA.CHART_DATA.
NOTE: PROCEDURE SGRENDER used (Total process time):
      real time          0.29 seconds
      cpu time           0.09 seconds
```

Default path: d:\Program Files\SAS\SASFoundation\9.2

ODS Graphics Designer

Chart looks good except for the vertical axis on the Series chart.



ODS Graphics Designer – Customizing the Output

Create yaxisopts for the boxplot and series plot overlays.

```
layout overlay _id='overlay' / walldisplay=(FILL)
   xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10
            family='Albany AMT') display=(TICKS TICKVALUES LINE ));
   yaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10
            family='Albany AMT')
            display=(TICKS TICKVALUES LINE )
            labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')
            label=_YLABEL);
    boxplot _id='box' x=_MONTH y=_YVARBOX / name='box';
 endlayout;

layout overlay _id='overlay2' / walldisplay=(FILL)
   xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10
            family='Albany AMT') display=(TICKS TICKVALUES LINE ))
   yaxisopts=(display=none);
```

Delete the rowaxes statements.

```
rowaxes;
    rowaxis _id='rowaxis' / label=_YLABEL
    labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')
    tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT');
endrowaxes;
```

ODS Graphics Designer

The finished chart as a PNG file.

How do I share the produced graph with others?



PDF Advantages

- ***Multiple charts available in a single file***
- ***More than one chart can be displayed on a page***
- ***Charts can be click, copy and paste into WORD or PowerPoint !!!***
- ***Charts can be part of a comprehensive report***

ODS Graphics Designer – PDF Output

ODS graphics on and specify the height and width of the output.

No need to set goptions device=sasprtc;

With ODS noresults there is no need for object= on the PROC SGRENDER line.

```
options orientation=landscape nonumber nodate;
 ods escapechar='^';
 ods listing image_dpi=100 close;
 ods noresults;
 ods pdf file="e:\temporary\file.pdf" notoc columns=2 startpage=no;

 /* Output is 4 graphs to a page in a 2x2 layout with an ods startpage
after each column */
   ods graphics on /height=3.3in width=4.9in;

    proc sgrender data=data.chart_data template=sg_ghsug;
      dynamic _DATE="date" _MONTH="month"
              _MEAN="mean" _MAX="max" _MIN="min"
              _YVARBOX="flow" _YLABEL="L/min"
              _TITLE="Flow" _FOOTNOTE="My SG chart";
    run;

    … more charts

 ods pdf close;
 ods listing;
 ods results;
```

ODS Graphics Designer – PDF Output

Effect of Output Size on Font Appearance

Default graph size produced for 8.5 x 11 paper in portrait mode



ODS Graphics Designer – PDF Output

Effect of Output Size on Font Appearance

Graph size 3.3in high 4.9in wide

Font height is absolute points. Font appears larger compared to the chart size.



ODS Graphics Designer – References

Using the ODS Graphics Designer to Create Your Own Templates. Philip R Holland. Paper 034-2010, SAS Global Forum 2010

ODS Graphics Designer An Interactive Tool for Creating Batchable Graphs. Sanjay Matange. NESUG 2009 Poster

SAS/GRAPH® 9.2: ODS Graphics Designer Help.

What's New in SAS 9.3 ODS Graphics Designer.

Support.sas.com

QUESTIONS?


THIS COMPLETES MY PRESENTATION.
DOES ANYONE HAVE A QUESTION DESIGNED TO INCREASE MY WORKLOAD FOR YOUR ENTERTAINMENT?
HOW MUCH MONEY WOULD THE COMPANY SAVE IF YOU DID THE ENTIRE PROJECT BY YOURSELF?
HMMM...
www.dilbert.com scottadams@aol.com
5-10-07 © 2007 Scott Adams, Inc./Dist. by UFS, Inc.